
6 Библиотеки Sys и Proc_Lib
The module sys contains functions for simple debugging of processes implemented using behaviours. 
Модуль sys содержит функции для простейшей отладки процессов, реализованных с использованием поведений. 

There are also functions that, together with functions in the module proc_lib, can be used to implement a special process, a process which comply to the OTP design principles without making use of a standard behaviour. They can also be used to implement user defined (non-standard) behaviours. 

Существуют также такие функции, которые совместно с функциями в модуле proc_lib, могут быть использованы для реализации специальных процессов (special process), процессов, которые соответсвтуют принципам дизайна OTP, но при этом не используют стандартных поведений. Они могут использоваться для реализации пользовательских(нестандартных) поведений. 

Both sys and proc_lib belong to the STDLIB application. 
И sys, и proc_lib являются частью приложения STDLIB. 

6.1 Simple Debugging
6.1 Простая отладка
The module sys contains some functions for simple debugging of processes implemented using behaviours. We use the code_lock example from the gen_event chapter to illustrate this: 
Модуль sys содержит некоторые функции для простой отладки процессов, реализованных с использованием поведений. Мы используем пример code_lock из главы gen_event, чтобы это продемонстрировать: 

% erl

Erlang (BEAM) emulator version 5.2.3.6 [hipe] [threads:0]

Eshell V5.2.3.6  (abort with ^G)

1> code_lock:start_link([1,2,3,4]).

{ok,<0.32.0>}

2> sys:statistics(code_lock, true).

ok

3> sys:trace(code_lock, true).

ok

4> code_lock:button(4).

*DBG* code_lock got event {button,4} in state closed

ok

*DBG* code_lock switched to state closed

5>  code_lock:button(3).

*DBG* code_lock got event {button,3} in state closed

ok

*DBG* code_lock switched to state closed

6> code_lock:button(2).

*DBG* code_lock got event {button,2} in state closed

ok

*DBG* code_lock switched to state closed

7> code_lock:button(1).

*DBG* code_lock got event {button,1} in state closed

ok

OPEN DOOR

*DBG* code_lock switched to state open

*DBG* code_lock got event timeout in state open

CLOSE DOOR

*DBG* code_lock switched to state closed

8> sys:statistics(code_lock, get). 

{ok,[{start_time,{{2003,6,12},{14,11,40}}},

     {current_time,{{2003,6,12},{14,12,14}}},

     {reductions,333},

     {messages_in,5},

     {messages_out,0}]}

9> sys:statistics(code_lock, false).

ok

10> sys:trace(code_lock, false).     

ok

11> sys:get_status(code_lock).

{status,<0.32.0>,

        {module,gen_fsm},

        [[{'$ancestors',[<0.30.0>]},

          {'$initial_call',{gen,init_it,

                                [gen_fsm,

                                 <0.30.0>,

                                 <0.30.0>,

                                 {local,code_lock},

                                 code_lock,

                                 [1,2,3,4],

                                 []]}}],

         running,

         <0.30.0>,

         [],

         [code_lock,closed,{[],[1,2,3,4]},code_lock,infinity]]}

6.2 Special Processes
6.2 Специальные процессы
This section describes how to write a process which comply to the OTP design principles, without making use of a standard behaviour. Such a process should: 

· be started in a way that makes the process fit into a supervision tree, 

· support the sys debug facilities, and 

· take care of system messages. 

Этот раздел описывает создание процессов, соответствующих принципам дизайна OTP, без использования стандартных поведений. Такие процессы должны: 

· запускаться таким образом, чтобы процесс стал частью дерева контроля, 

· поддерживать sys отладочные возможности(debug facilities), и 

· позаботиться об обработке системных сообщений(system messages). 

System messages are messages with special meaning, used in the supervision tree. Typical system messages are requests for trace output, and requests to suspend or resume process execution (used during release handling). Processes implemented using standard behaviours automatically understand these messages. 
Системные сообщения – это сообщения имеющие специальное назначение, используемые в дереве контроля. Типичные системные сообщения это запросы на вывод данных, а также запросы на временную остановку или повторный запуск процессов (используемые при работе с релизами). Процессы, реализованные с помощью стандартных поведений, автоматически понимают эти сообщения.
6.2.1 Example
6.2.1 Пример
The simple server from the Overview chapter, implemented using sys and proc_lib so it fits into a supervision tree: 
Простой сервер из главы «Обозрение» Overview, реализованный с использованием sys и proc_lib , таким образом он попадает в дерево контроля: 

-module(ch4).

-export([start_link/0]).

-export([alloc/0, free/1]).

-export([init/1]).

-export([system_continue/3, system_terminate/4,

         write_debug/3]).

start_link() ->

    proc_lib:start_link(ch4, init, [self()]).

alloc() ->

    ch4 ! {self(), alloc},

    receive

        {ch4, Res} ->

            Res

    end.

free(Ch) ->

    ch4 ! {free, Ch},

    ok.

init(Parent) ->

    register(ch4, self()),

    Chs = channels(),

    Deb = sys:debug_options([]),

    proc_lib:init_ack(Parent, {ok, self()}),

    loop(Chs, Parent, Deb).

loop(Chs, Parent, Deb) ->

    receive

        {From, alloc} ->

            Deb2 = sys:handle_debug(Deb, {ch4, write_debug},

                                    ch4, {in, alloc, From}),

            {Ch, Chs2} = alloc(Chs),

            From ! {ch4, Ch},

            Deb3 = sys:handle_debug(Deb2, {ch4, write_debug},

                                    ch4, {out, {ch4, Ch}, From}),

            loop(Chs2, Parent, Deb3);

        {free, Ch} ->

            Deb2 = sys:handle_debug(Deb, {ch4, write_debug},

                                    ch4, {in, {free, Ch}}),

            Chs2 = free(Ch, Chs),

            loop(Chs2, Parent, Deb2);

        {system, From, Request} ->

            sys:handle_system_msg(Request, From, Parent,

                                  ch4, Deb, Chs)

    end.

system_continue(Parent, Deb, Chs) ->

    loop(Chs, Parent, Deb).

system_terminate(Reason, Parent, Deb, Chs) ->

    exit(Reason).

write_debug(Dev, Event, Name) ->

    io:format(Dev, "~p event = ~p~n", [Name, Event]).

Example on how the simple debugging functions in sys can be used for ch4 as well: 
Пример использования простых отладочных функций может быть использоват также для ch4 : 

% erl

Erlang (BEAM) emulator version 5.2.3.6 [hipe] [threads:0]

Eshell V5.2.3.6  (abort with ^G)

1> ch4:start_link().

{ok,<0.30.0>}

2> sys:statistics(ch4, true).

ok

3> sys:trace(ch4, true).

ok

4> ch4:alloc().

ch4 event = {in,alloc,<0.25.0>}

ch4 event = {out,{ch4,ch1},<0.25.0>}

ch1

5> ch4:free(ch1).

ch4 event = {in,{free,ch1}}

ok

6> sys:statistics(ch4, get).

{ok,[{start_time,{{2003,6,13},{9,47,5}}},

     {current_time,{{2003,6,13},{9,47,56}}},

     {reductions,109},

     {messages_in,2},

     {messages_out,1}]}

7> sys:statistics(ch4, false).

ok

8> sys:trace(ch4, false).

ok

9> sys:get_status(ch4).

{status,<0.30.0>,

        {module,ch4},

        [[{'$ancestors',[<0.25.0>]},{'$initial_call',{ch4,init,[<0.25.0>]}}],

         running,

         <0.25.0>,

         [],

         [ch1,ch2,ch3]]}

6.2.2 Starting the Process
6.2.2 Запуск процессов
A function in the proc_lib module should be used to start the process. There are several possible functions, for example spawn_link/3,4 for asynchronous start and start_link/3,4,5 for synchronous start. 
Для запуска процесса должна быть использована функция из модуля proc_lib Есть несколько функций подходящих для этой цели, например spawn_link/3,4 для асинхронного запуска и start_link/3,4,5 для синхронного.

A process started using one of these functions will store information that is needed for a process in a supervision tree, for example about the ancestors and initial call. 
Процесс запускается используя одну из этих функций, для сохранения информации необходимой для процесса в дереве контроля, например о потомках и первом вызове.
Also, if the process terminates with another reason than normal or shutdown, a crash report (see SASL User's Guide) is generated. 

In the example, synchronous start is used. The process is started by calling ch4:start_link(): 
Например используется синхронный запуск. Процесс запускается вызовом ch4:start_link():

start_link() ->

    proc_lib:start_link(ch4, init, [self()]).

ch4:start_link calls the function proc_lib:start_link. This function takes a module name, a function name and an argument list as arguments and spawns and links to a new process. The new process starts by executing the given function, in this case ch4:init(Pid), where Pid is the pid (self()) of the first process, that is the parent process. 
ch4:start_link вызывает функцию proc_lib:start_link. Эта функция получает в качестве аргументов имя модуля, имя функции и список аргументов, создает новый процесс и устанавливает связь с ним. Новый процесс запускается вызовом заданной функции, в нашем случае ch4:init(Pid), где Pid это  pid (self()) первого процесса, который является родительским процессом. 

In init, all initialization including name registration is done. The new process must also acknowledge that it has been started to the parent: 
В функции init,производится вся инициализация, включая регистрацию имению Новый процесс должен также уведомить родителя о своем успешном старте.
init(Parent) ->

    ...

    proc_lib:init_ack(Parent, {ok, self()}),

    loop(...).

proc_lib:start_link is synchronous and does not return until proc_lib:init_ack has been called. 
proc_lib:start_link – синхронная функция, и не вернет управление до тех пор, пока не будет вызвана proc_lib:init_ack. 
6.2.3 Debugging
6.2.3 Отладка
To support the debug facilites in sys, we need a debug structure, a term Deb which is initialized using sys:debug_options/1: 
Для поддержки отладочных возможностей в sys, нам нужна отладочная структура(debug structure), терм Deb, который инициализируется с помощью функции sys:debug_options/1: 

init(Parent) ->

    ...

    Deb = sys:debug_options([]),

    ...

    loop(Chs, Parent, Deb).

sys:debug_options/1 takes a list of options as argument. Here the list is empty, which means no debugging is enabled initially. See sys(3) for information about possible options. 
sys:debug_options/1 принимает список опций в качестве аргумента. Если список пуст, это означает, что отладка не включена. Смотрите документацию по sys(3) для получения информации о возможных опциях. 

Then for each system event that we want to be logged or traced, the following function should be called. 
После этого, для каждого системного события, которое мы хотим залогировать или оттрасировать, должна быть вызвана следующая функция :
sys:handle_debug(Deb, Func, Info, Event) => Deb1

· Deb is the debug structure.

· Func is a tuple {Module, Name} (or a fun) and should specify a (user defined) function used to format trace output. For each system event, the format function is called as Module:Name(Dev, Event, Info), where:

· Dev is the IO device to which the output should be printed. See io(3).

· Event and Info are passed as-is from handle_debug.

· Info is used to pass additional information to Func, it can be any term and is passed as-is.

· Event is the system event. It is up to the user to define what a system event is and how it should be represented, but typically at least incoming and outgoing messages are considered system events and represented by the tuples {in,Msg[,From]} and {out,Msg,To}, respectively.

· Deb это отладочная структура.

· Func это кортеж {Module, Name} (или функция) и должен указывать (определенную пользователем) функцию для форматирования трассировочных данных. Для каждого системного события, функция форматирования вызывается так : Module:Name(Dev, Event, Info), где:

· Dev устройство ввода-вывода, на которое будет осуществлен вывод данных. См. io(3).

· Event и Info передаются в неизменяемом виде из handle_debug.

· Info использутся для передачи дополнительной информации в Func, это может быть любой терм, и он передается в первозданном виде.

· Event это системное событие. Пользователь определяет, чем является системное событие, и как оно должно быть представлено,но обычно, по крайней мере входящие и исходящие сообщения соответствуют системным сообщениям и представляются кортежами {in,Msg[,From]} и {out,Msg,To}, соответственно.

handle_debug returns an updated debug structure Deb1. 
handle_debug возвращает обновленную отладочную структуру Deb1. 

In the example, handle_debug is called for each incoming and outgoing message. The format function Func is the function ch4:write_debug/3 which prints the message using io:format/3. 
В этом примере handle_debug вызвается при каждом входящем и исходящем сообщениях. Функция форматирования Func это функция ch4:write_debug/3 , которая выводит сообщение используя io:format/3.
loop(Chs, Parent, Deb) ->

    receive

        {From, alloc} ->

            Deb2 = sys:handle_debug(Deb, {ch4, write_debug},

                                    ch4, {in, alloc, From}),

            {Ch, Chs2} = alloc(Chs),

            From ! {ch4, Ch},

            Deb3 = sys:handle_debug(Deb2, {ch4, write_debug},

                                    ch4, {out, {ch4, Ch}, From}),

            loop(Chs2, Parent, Deb3);

        {free, Ch} ->

            Deb2 = sys:handle_debug(Deb, {ch4, write_debug},

                                    ch4, {in, {free, Ch}}),

            Chs2 = free(Ch, Chs),

            loop(Chs2, Parent, Deb2);

        ...

    end.

write_debug(Dev, Event, Name) ->

    io:format(Dev, "~p событие = ~p~n", [Name, Event]).

6.2.4 Handling System Messages
6.2.4 Обработка системных сообщений
System messages are received as: 
Системное сообщение поступает в таком виде :
{system, From, Request}

The content and meaning of these messages do not need to be interpreted by the process. Instead the following function should be called: 
Содержимое и смысл этих сообщений не должны как-либо обрабатываться процессом. Вместо этого должна быть вызвана следующая функция :
sys:handle_system_msg(Request, From, Parent, Module, Deb, State)

This function does not return. It will handle the system message and then call: 
Эта функция не должна возвращать управление. Она обработает системное сообщение и затем вызовет :
Module:system_continue(Parent, Deb, State)

if process execution should continue, or: 
если процесс должен продолжать свою работу, или :
Module:system_terminate(Reason, Parent, Deb, State)

if the process should terminate. Note that a process in a supervision tree is expected to terminate with the same reason as its parent. 

· Request and From should be passed as-is from the system message to the call to handle_system_msg. 

· Parent is the pid of the parent. 

· Module is the name of the module. 

· Deb is the debug structure. 

· State is a term describing the internal state and is passed to system_continue/system_terminate. 

Если процесс должен завершить свою работу. Заметьте, что процесс в дереве контроля должен завершить свою работу по той же причине, что и родитель. 

· Request и From должны передаваться в неизменном виде от системного сообщения до вызова handle_system_msg. 

· Parent это pid родителя. 

· Module это имя модуля. 

· Deb это отладочная структура. 

· State это терм, описывающий внутреннее состояние, и он передается в system_continue/system_terminate. 

In the example: 
В примере :

loop(Chs, Parent, Deb) ->

    receive

        ...

        {system, From, Request} ->

            sys:handle_system_msg(Request, From, Parent,

                                  ch4, Deb, Chs)

    end.

system_continue(Parent, Deb, Chs) ->

    loop(Chs, Parent, Deb).

system_terminate(Reason, Parent, Deb, Chs) ->

    exit(Reason).

If the special process is set to trap exits, note that if the parent process terminates, the expected behavior is to terminate with the same reason: 
Если специальный процесс сконфигурирован так, чтобы отлавливать сообщения о завершении работы, необходимо заметить, что если родительский процесс завершает свою работу, ожидаемым поведением для процесса будет завершить свою работу с тем же сообщением.
init(...) ->

    ...,

    process_flag(trap_exit, true),

    ...,

    loop(...).

loop(...) ->

    receive

        ...

        {'EXIT', Parent, Reason} ->

            ..maybe some cleaning up here..

            exit(Reason);

        ...

    end.

6.3 User-Defined Behaviours
6.3 Пользовательские поведения
To implement a user-defined behaviour, write code similar to code for a special process but calling functions in a callback module for handling specific tasks. 
Для реализации пользовательского поведения, напишите код, подобный коду специального процесса, но вызывающий функции из модуля обратного вызова для выполнения определенных действий.
If it is desired that the compiler should warn for missing callback functions, as it does for the OTP behaviours, implement and export the function: 
Если вы хотите, чтобы компилятор предупреждал об отсутствии функций обратного вызова, точно так же, как для поведений OTP, реализуйте и экспортируйте следующую функцию :
behaviour_info(callbacks) ->

    [{Name1,Arity1},...,{NameN,ArityN}].

where each {Name,Arity} specifies the name and arity of a callback function. 
Где каждый из кортежей {Name,Arity} определяет имя и арность функции обратного вызова.
When the compiler encounters the module attribute -behaviour(Behaviour). in a module Mod, it will call Behaviour:behaviour_info(callbacks) and compare the result with the set of functions actually exported from Mod, and issue a warning if any callback function is missing. 
Когда компилятор встречает атрибут модуля -behaviour(Behaviour) в модуле Mod , он вызывает Behaviour:behaviour_info(callbacks) и сравнивает результта с множеством функций, экспортированных из , и сообщает об ошибке, силе какая-либо функция обратного вызова отсутствует.
Example: 
Пример: 
%% User-defined behaviour module

-module(simple_server).

-export([start_link/2,...]).

-export([behaviour_info/1]).

behaviour_info(callbacks) ->

    [{init,1},

     {handle_req,1},

     {terminate,0}].

start_link(Name, Module) ->

    proc_lib:start_link(?MODULE, init, [self(), Name, Module]).

init(Parent, Name, Module) ->

    register(Name, self()),

    ...,

    Dbg = sys:debug_options([]),

    proc_lib:init_ack(Parent, {ok, self()}),

    loop(Parent, Module, Deb, ...).

...

In a callback module: 
В модуле обратного вызова:
-module(db).

-behaviour(simple_server).

-export([init/0, handle_req/1, terminate/0]).







